

# Interpolazione Polinomiale metodo di Lagrange

Enrico Bertolazzi

```
> # carica la libreria per l'algebra lineare  
with(LinearAlgebra):  
> # definisco i dati del problema  
# N = numero di punti  
# X = coordinate x dei punti da interpolare  
# Y = coordinate y dei punti da interpolare  
N,X,Y := 4, <1,3,5/2,4>, <2,2,4,4> ;
```

$$N, X, Y := 4, \begin{bmatrix} 1 \\ 3 \\ \frac{5}{2} \\ 4 \end{bmatrix}, \begin{bmatrix} 2 \\ 2 \\ 4 \\ 4 \end{bmatrix} \quad (1)$$

```
> # F(k) definira` il polinomio F per il k-esimimo punto  
F := k -> mul(x-X[j],j=1..N)/(x-X[k]);
```

$$F := k \rightarrow \frac{\text{mul}(x - X_j, j = 1..N)}{x - X_k} \quad (2)$$

```
> # alcuni esempi  
F(1); F(2);
```

$$(x-3) \left(x - \frac{5}{2}\right) (x-4) \quad (3)$$
$$(x-1) \left(x - \frac{5}{2}\right) (x-4)$$

```
> # L(k) definira` il polinomio L per il k-esimimo punto  
L := k -> F(k)/subs(x=X[k],F(k));
```

$$L := k \rightarrow \frac{F(k)}{\text{subs}(x = X_k, F(k))} \quad (4)$$

```
> # alcuni esempi  
L(1); L(2);
```

$$-\frac{1}{9} (x-3) \left(x - \frac{5}{2}\right) (x-4) \quad (5)$$
$$-(x-1) \left(x - \frac{5}{2}\right) (x-4)$$

```
> # costruzione del polinomio interpolante
```

```
PL := add(Y[i]*L(i),i=1..N) ;
```

$$PL := -\frac{2}{9}(x-3)\left(x-\frac{5}{2}\right)(x-4) - 2(x-1)\left(x-\frac{5}{2}\right)(x-4) + \frac{32}{9}(x-1)(x-3)(x-4) + \frac{8}{9}(x-1)(x-3)\left(x-\frac{5}{2}\right)$$

**(6)**

```
> # semplificazione del polinomio
PL := simplify(PL);
```

$$PL := -\frac{68}{3} + \frac{356}{9}x - \frac{154}{9}x^2 + \frac{20}{9}x^3$$

**(7)**

```
> # polottaggio del polinomio interpolante
A := plot(PL,x=0.5..4.5,thickness=4):
# punti di interpolazione
B := plot([seq([X[i],Y[i]],i=1..N)],
           style=POINT,
           color=blue,
           symbolsize=50):
# punti proiettati sull'asse X e Y=1
C := plot([seq([X[i],0],i=1..4),
            seq([X[i],1],i=1..4)],
           style=POINT,
           color=black,
           symbol=CIRCLE,
           symbolsize=30):
# plottaggio dei polinomi L(k)
E := plot([seq(L(i),i=1..N)],x=0..5,thickness=2):
> display({A,B,C,E});
```

